

TECHNICAL REFERENCE

Power Apps: concepts & logic

Every control, formula, and Power Fx pattern behind the Weekly Escalation Tracker canvas app

The pattern that made multi-value lookups work:

```
ForAll(cmb.SelectedItems, { Id: ID, Value: Text('Unique ID') })
```

Canvas app · Power Fx · SharePoint · Responsive layout

Siddhant Singh, Senior Data Analyst & Power Platform Developer

APP SETUP

The design system and app state

The app is a responsive canvas app. Its theme colours and the active-view flag are defined once in `App.OnStart` and reused everywhere, so styling lives in one place. `OnStart` does not run automatically in the editor: after editing it, right-click **App** and choose **Run OnStart**.

```
// App > OnStart    (run manually: right-click App > Run OnStart)
Set(gPrimary,      ColorValue("#4F46E5"));
Set(gPrimaryDark,  ColorValue("#4338CA"));
Set(gBg,           ColorValue("#F5F6FA"));
Set(gCard,         White);
Set(gText,         ColorValue("#1F2430"));
Set(gMuted,        ColorValue("#6B7280"));
Set(gSuccess,      ColorValue("#16A34A"));
Set(gDanger,       ColorValue("#DC2626"));
Set(gWarning,      ColorValue("#F59E0B"));
Set(gView, "about")    // which Home tab is active
```

`Set()` creates a **global** variable (readable on every screen). For state that only one screen needs, `UpdateContext({ ... })` keeps it local to that screen.

HOME SCREEN

Two views from a single variable

The About and Meeting views are one screen, not two. A single global variable, `gView`, decides which gallery shows. The tab buttons only set it; each gallery's `Visible` compares it.

```
// btn_About > OnSelect
Set(gView, "about")

// btn_Meeting > OnSelect
Set(gView, "meeting")

// gal_About > Visible
gView = "about"

// gal_Meeting > Visible
gView = "meeting"
```

Column headers sit on the **screen**, not inside a gallery (a gallery repeats its contents for every row). They line up with the columns by referencing the gallery and label positions, for example `gal_About.X + lbl_Client.X`, and share the same `Visible` expression.

HOME GALLERY

Search and sort on the list

The gallery filters by the search box and sorts newest-first. Two details make it work: a text input exposes `.Text` (not `.Value`), and you **cannot** filter inside a multi-value lookup, so the search targets a plain-text column with the `in` (contains) operator.

```
// gal_About > Items      (gal_Meeting uses the same formula)
Sort(
    Filter(
        'Power Apps EXP_ Siddhant',
        IsBlank('TXT Search'.Text)
        || 'TXT Search'.Text in 'Utility Client Name'
    ),
    ID, SortOrder.Descending
)
```

When the box is empty, `IsBlank( . . . )` is true and every row shows; otherwise only rows whose client name contains the typed text remain.

DISPLAYING DATA

Labels and choice pills

SharePoint's complex columns come back in different shapes. A multi-value lookup is a **table**, so it is flattened with `Concat`. A single-select choice is a **record**, so `.Value` gives its text, and `Switch` maps each value to a colour.

```
// lbl_UniqueID > Text      (Unique ID is a multi-value lookup = a table)
Concat(ThisItem.'Unique ID', Text(Round(Value, 0)), "; ")

// lbl_Client > Text        (plain-text column, filled by the flow)
ThisItem.'Utility Client Name'

// pill_Status > Text
ThisItem.Status.Value

// pill_Status > Fill
Switch(
    ThisItem.Status.Value,
    "Open",      ColorValue("#FEF3C7"),
    "On time",   ColorValue("#166534"),
    "At Risk",   ColorValue("#991B1B"),
    "Completed", ColorValue("#DCFCE7"),
    gCard
)

// pill_Slide > Fill        (Yes/No choice)
If(ThisItem.'Included in Project Slide'.Value = "Yes", gSuccess, gDanger)
```

Editor note. On ThisItem formulas you may see “This formula uses scope, which is not presently supported for evaluation.” That message is **informational** (the editor has no row context), not an error. Press **F5** and it renders correctly.

NEW ENTRY

Pick once, auto-fill the rest

The combo box is bound to the master list with multi-select on, so its `SelectedItems` is a table. Four preview labels read those selected rows live, again with `Concat`. Dropdowns are populated by `Choices()`, and the two toggles capture the yes/no flags.

```
// cmb_UniqueID (properties)
Items = 'Mapping Table'
SelectMultiple = true // "Allow multiple selections" = On
SearchFields = ["field_0"] // field_0 = internal name of "Unique ID"
DisplayFields = ["field_0"]

// lbl_Client_View > Text (live preview from the picked rows)
Concat(cmb_UniqueID.SelectedItems, 'Utility Client Name', "; ")

// drp_Status > Items
Choices('Power Apps EXP_ Siddhant'.Status)

// drp_Risk > Items
Choices('Power Apps EXP_ Siddhant'.Project Risk Category')
```

THE SAVE

One Patch, three column types

Submit writes a new row with a single Patch. The important part is shaping each complex type correctly: a multi-value lookup needs ForAll to build { Id, Value } records (and the numeric ID is wrapped with Text()); a choice takes the whole .Selected record; a yes/no choice takes { Value: "Yes" }. A saving flag drives the spinner, and the form resets on success.

```
// btn_Submit > OnSelect
Set(var_Saving, true);

Patch(
    'Power Apps EXP_ Siddhant',
    Defaults('Power Apps EXP_ Siddhant'),
    {
        'Unique ID':
            ForAll(cmb_UniqueID.SelectedItems,
                { Id: ID, Value: Text('Unique ID') } ),
        Status: drp_Status.Selected,
        'Project Risk Category': drp_Risk.Selected,
        'Included in Project Slide':
            { Value: If(tgl_Slide.Value, "Yes", "No") },
        'Included in Escalation Meeting':
            { Value: If(tgl_Escalation.Value, "Yes", "No") }
    }
);

Set(var_Saving, false);
Notify("Entry saved", NotificationType.Success);
Reset(cmb_UniqueID);
Navigate(scrHome)
```

Supporting bits: `spinner.Visible = var_Saving` shows the loading state, and an empty-state label uses `CountRows(gal_About.AllItems) = 0`.

PATTERNS & GOTCHAS

The rules worth remembering

- **Multi-value lookup: table in, table out.** Display with `Concat(...)`; save with `ForAll(..., { Id, Value })`.
- **Choice column.** Save the whole record with `.Selected`; display the text with `.Value`.
- **Yes/No choice.** Save as `{ Value: "Yes" }` or `{ Value: "No" }`, not a bare boolean.
- **Text input.** Read `.Text`, never `.Value`.
- **Search.** You cannot `Filter` inside a lookup (a scope error). Filter a plain-text column with `in`.
- **Internal names differ from display names.** "Unique ID" is `field_0`; type `ThisItem`. and pick from the dropdown.
- **Global vs screen state.** `Set()` is app-wide; `UpdateContext()` is screen-only.
- **OnStart.** It does not auto-run in the editor, right-click **App > Run OnStart** after edits.
- **"Scope not supported" on ThisItem.** Informational only; confirm with **F5**.

FUNCTION GLOSSARY

The Power Fx used here, in one place

Patch	Create or update a record in a data source.
ForAll	Loop a table and return a value (here, reshape lookup rows to records).
Filter	Return the rows of a table that meet a condition.
Search / in	Text-contains matching; "x in col" is true when col contains x.
Sort	Order a table by a column and direction (<code>SortOrder.Descending</code>).
Lookup	Return the first record that matches a condition.
Choices	Return the allowed values of a choice or lookup column.
Concat	Join a table's values into one string with a separator.
Set	Create or set a global variable (app-wide).
UpdateContext	Create or set a screen-scoped variable.

Navigate	Move to another screen.
Reset	Clear a control back to its default.
Notify	Show a banner message to the user.
IsBlank	True when a value is empty; used to detect an empty search box.
CountRows	Count the rows of a table; used for the empty state.
Switch / If	Choose a result from several conditions (pill colours, flags).
Text / Round	Format a number as text; round it first where needed.
ColorValue	Turn a hex string like "#4F46E5" into a colour value.

Reference companion to the Weekly Escalation Tracker case study

Every formula above is the exact Power Fx used in the app, documented so it can be read, reused, or explained end to end.